

**EJEMPLO 7**

Dados los límites inferior y superior de 10 subintervalos (clases) así como la frecuencia absoluta para cada uno, calcular:

- ☞ Frecuencia absoluta acumulada para cada subintervalo.
- ☞ Frecuencia relativa para cada subintervalo.
- ☞ Marca de clase para cada subintervalo.
- ☞ La media de los datos.

Supongamos que los datos son las edades de los alumnos de FACIJUP – ULA y que han sido agrupados en las siguientes clases:

Edades		Número de estudiantes
15	20	1345
20	25	1767
25	30	567
30	35	345
25	40	132
40	45	93
45	50	64
50	55	39
55	60	12
60	65	9

Cada intervalo posee un valor mínimo y uno máximo; y en la celda correspondiente al número de estudiantes se encuentra; la cantidad de personas que tienen la edad comprendida entre esos dos valores, llamada frecuencia absoluta del número de observaciones y signado por  $f_i$ .

Primero se explicará someramente el cálculo involucrado y luego simplemente se mostrará como deberá ser la apariencia aproximada final de la hoja de cálculo, ya que solo hay cálculos involucrados que deberá hacer como repaso de lo visto hasta el momento; para solo detenernos un poco en la parte del gráfico.

La **frecuencia absoluta acumulada** de un subintervalo, es la suma de las frecuencias absolutas hasta ese intervalo, es decir todos los valores anteriores, signada por  $F_i$ . De hacer los cálculos correctamente, en el último sub-intervalo la frecuencia absoluta

acumulada arrojará el total de las observaciones, signado en estadística como  $n$ .

La **frecuencia relativa** de un subintervalo, es el cociente de dividir la frecuencia absoluta ( $f_i$ ) de ese subintervalo entre el total de observaciones ( $n$ ).

Para obtener la **Marca de clase** ( $M_i$ ), se debe sumar los límites del subintervalo y dividirlos entre dos, es algo así como el punto medio de la clase.

Este ejemplo es realmente sencillo y lo primero es vaciar la información dada de una manera apropiada que permita la comprensión de la información y facilite los cálculos, en este orden la

**Frecuencia Absoluta**

$$F_i = \sum_{i=1}^{\#intervalos} f_i$$

**Frecuencia Relativa**

$$f_r = f_i / n$$


**Marca de Clase**

$$M_i = (Lim_{sup} + Lim_{inf}) / 2$$

experiencia me indica que éste es el principal error que se comete así que, comenzaremos por colocar cada uno de los sub-intervalos en dos columnas, es decir, el limite inferior del subintervalo en una celda y el limite superior en la otra celda como se muestra en la figura (de nuevo esto es solo un sugerencia, esta en la libertad de colocarlos de otra forma).

Para esto:

1. Coloque las etiquetas **Edades Sub-intervalo**, **Frecuencia Absoluta**, **Frecuencia Abs Acumulada**, **Frecuencia Relativa** y **Marca de Clase** en las celdas **B6**, **C6**, **D6**, **E6**, **F6** y **G6** respectivamente.



**Figura 26** Formato de celda. Control de apariencia, orientación y alineación del texto dentro de la(s) celda(s)

Hasta acá su hoja de cálculo tendrá la apariencia que no es muy prolija, esto se puede solucionar dándole un formato apropiado a las celdas, tal y como se mencionó en la sección de formatos; sin embargo; ampliaremos un poco en este problema.

Seleccione el rango desde A6 hasta F6, luego del menú formato seleccione la opción celda, luego sobre el cuadro de dialogo formato de celda (Figura 26) active la ficha alineación. Active el cuadro de activación **Ajustar texto** en la **sección Control de texto**, seleccione **centrar** en las listas desplegables horizontal y vertical de la **sección Alineación del texto**. Todo esto en la misma ficha, note que incluso se puede variar la orientación (figura 26). Anímese a indagar y cambiar los bordes y tramas de estas celdas.

Seleccione las celdas A6 y B6, de seguidas de un clic en el **botón combinar y centrar** o acceda de nuevo al menú **f**ormato, opción **c**eldas y a la ficha alineación y active combinar celdas, observe atentamente el efecto.



- Coloque las abreviaturas **Inf** e la celda **A7**, **Sup** en la celda **B7**, **f** en la **D7**, **F** en **E7**, **fr** en **E7** y **M** en **F7**

Para el primer cálculo de la frecuencia absoluta acumulada del primer sub-intervalo, como no hay valores anteriores será simplemente el valor de la frecuencia absoluta de ese subintervalo, es decir la fórmula será simplemente **=C8**, para el segundo y sucesivos será, la suma de los valores de la frecuencia absoluta del subintervalo más la de los sub-intervalos anteriores a él, pero si observa estos están en la celda anterior de la frecuencia absoluta acumulada, por lo que la formula para el segundo será, **= C9 + D8**. Y para los demás bastará con arrastrar el cuadro de relleno como lo hemos hecho en los ejemplos anteriores. Note que en el último subintervalo de la frecuencia absoluta acumulada (celda **D17**) tendremos el total de las observaciones (**n**).

Para la frecuencia relativa, basta situarnos en la celda **E8** que es el valor para el primer sub-intervalo y teclear la formula **=C8/D\$17** para obtener la frecuencia relativa para ese sub-intervalo, para los demás basta arrastrar el cuadro de relleno de esa celda. Note que la referencia al valor de **n** (celda **D17**) es una referencia absoluta a la fila.

Para la marca de clase basta con teclear la formula **=(A8+B8)/2**, en la celda correspondiente y luego completar para todos los sub-intervalos mediante el cuadro de relleno, una observación en este cálculo es el uso de los paréntesis, para alterar el orden de precedencia (orden de evaluación) ya que, la división tiene mayor precedencia que la suma por lo que se evaluaría primero arrojando un valor errado, e decir sumaria a **A8** el resultado de **B8/2**, que obviamente estaría mal.

Menú **f**ormato opción **c**eldas



**Botón combinar y centrar;** para combinar dos celdas en una sola y compartan el mismo texto.

Para la media una opción es en una columna adicional colocar el producto de la frecuencia absoluta por la marca de clase de cada intervalo, par luego en otra celda sumar esta columna y dividirlo entre el total de observaciones **n** (celda **D17**), el contenido final de su hoja en cuanto a formulas deberá ser similar al que se muestra en la figura 27 de seguidas.

La otra opción para la media es emplear la función SUMAPRODUCTO, indague como utilizarla haciendo uso del asistente para funciones que ya se ha usado ampliamente en los problemas anteriores.

	A	B	C	D	E	F	G	H	I
1									
2									
3									
4									
5									
6	Edad Subintervalo		Frecuencia Absoluta	Frecuencia Abs Acumulada	Frecuencia Relativa	Marca de Clase	Marca * Frec Abs	Media	Media
7	Inf	Sup	fi	Fi	fr	M	fi * M		
8	15	20	1345	=C8	=C8/D\$17	=(A8+B8)/2	=C8*F8	=SUMA(G8:G17)/D17	=SUMAPRODUCTO(C8:C17,F8:F17)/D17
9	20	25	1767	=C9+D8	=C9/D\$17	=(A9+B9)/2	=C9*F9		
10	25	30	567	=C10+D9	=C10/D\$17	=(A10+B10)/2	=C10*F10		
11	30	35	345	=C11+D10	=C11/D\$17	=(A11+B11)/2	=C11*F11		
12	35	40	132	=C12+D11	=C12/D\$17	=(A12+B12)/2	=C12*F12		
13	40	45	93	=C13+D12	=C13/D\$17	=(A13+B13)/2	=C13*F13		
14	45	50	64	=C14+D13	=C14/D\$17	=(A14+B14)/2	=C14*F14		
15	50	55	39	=C15+D14	=C15/D\$17	=(A15+B15)/2	=C15*F15		
16	55	60	12	=C16+D15	=C16/D\$17	=(A16+B16)/2	=C16*F16		
17	60	65	9	=C17+D16	=C17/D\$17	=(A17+B17)/2	=C17*F17		

Figura 27. Hoja final para el ejemplo 6. Estadística para datos agrupados

La apariencia anterior la puede lograr activando la opción *formulas* en la ficha ver del cuadro de dialogo opciones, que se encuentra en el menú herramientas. Esta apariencia es solo demostrativa pues lo importante es obtener los cálculos.